



Integrated Circuits Design by FPGA

م.م. أحمد مؤيد عبدالحسين
جامعة الفرات الأوسط التقنية / الكلية التقنية الهندسية / نجف

Lecture 9

Finite State Machines (FSM)

Traffic Light Controller (TLC)

Objectives of this Lecture


- To implement the example of **Traffic Light Controller (TLC)**, using **FSM** method by FPGA technique.
- To understand the **FSM** method.

Contents of this Lecture

- **FSM** introduction. (Review)
- **TLC** example

FSM introduction

- **FSM** can be very helpful in the design of certain types of systems, particularly those whose tasks form a well-defined sequence (digital controllers, for example).



FSM introduction

- Figure 8.1 shows the block diagram of a single-phase state machine. As indicated in the figure, the lower section contains the sequential logic (flip-flops), while the upper section contains the combinational logic (parallel VHDL).




Figure 8.1
Mealy (Moore) state machine diagram.

FSM introduction

- The combinational (upper) section has two inputs, being one **pr_state** (present state) and the other the external input proper. It has also two outputs, **nx_state** (next state) and the external output proper.




Figure 8.1
Mealy (Moore) state machine diagram.

FSM introduction

- The sequential (lower) section has three inputs (clock, reset, and nx_state), and one output (pr_state). Since all flip-flops are in this part of the system, clock and reset must be connected to it.




Figure 8.1
Mealy (Moore) state machine diagram.

FSM introduction

- From a VHDL perspective, it is clear that the lower part, being sequential, will require a PROCESS, while the upper part, being combinational, will not.
- However, it is also possible to use PROCESS inside the upper part to implement combinational (parallel) design.





Figure 8.1
Mealy (Moore) state machine diagram.

FSM introduction

- One important aspect related to the **FSM** approach is that, though any sequential circuit can in principle be modeled as a state machine, this is not always advantageous. The reason is that the code might become longer, more complex, and more error prone than in a conventional approach. This is often the case with simple registered circuits, like counters.
- The **FSM** approach is advisable in systems whose tasks constitute a well - structured list so all states can be easily enumerated. That is, in a typical state machine implementation, we will encounter, at the beginning of the ARCHITECTURE, a user-defined enumerated data type, containing a list of all possible system states.

TLC Example



State	Operation Mode		
	REGULAR	TEST	STANDBY
	Time	Time	Time
RG	timeRG (30s)	timeTEST (1s)	---
RY	timeRY (5s)	timeTEST (1s)	---
GR	timeGR (45s)	timeTEST (1s)	---
YR	timeYR (5s)	timeTEST (1s)	---
YY	---	---	Indefinite

TLC Example





Figure 8.10
Specifications and states diagram (regular mode) for example 8.5.

TLC Example



TLC Example

```
1  -----
2  LIBRARY ieee;
3  USE ieee.std_logic_1164.all;
4  -----
5  ENTITY tlc IS
6      PORT ( clk, stby, test: IN STD_LOGIC;
7            r1, r2, y1, y2, g1, g2: OUT STD_LOGIC);
8  END tlc;
9  -----
10 ARCHITECTURE behavior OF tlc IS
11     CONSTANT timeMAX : INTEGER := 4500;    -- 45 sec
12     CONSTANT timeRG  : INTEGER := 3000;    -- 30 sec
13     CONSTANT timeRY  : INTEGER := 500;     -- 5 sec
14     CONSTANT timeGR  : INTEGER := 4500;    -- 45 sec
15     CONSTANT timeYR  : INTEGER := 500;     -- 5 sec
16     CONSTANT timeTEST : INTEGER := 100;    -- 1 sec
17     TYPE state IS (RG, RY, GR, YR, YY);
18     SIGNAL pr_state, nx_state: state;
19     SIGNAL time : INTEGER RANGE 0 TO timeMAX;
```



TLC Example

```
20 BEGIN
21     ----- Lower section of state machine: -----
22     PROCESS (clk, stby)
23         VARIABLE count : INTEGER RANGE 0 TO timeMAX;
24     BEGIN
25         IF (stby='1') THEN
26             pr_state <= YY;
27             count := 0;
28         ELSIF (clk'EVENT AND clk='1') THEN
29             count := count + 1;
30             IF (count = time) THEN
31                 pr_state <= nx_state;
32                 count := 0;
33             END IF;
34         END IF;
35     END PROCESS;
```



TLC Example

```
36 ----- Upper section of state machine: ----
37 PROCESS (pr_state, test)
38 BEGIN
39     CASE pr_state IS
40         WHEN RG =>
41             r1<='1'; r2<='0'; y1<='0'; y2<='0'; g1<='0'; g2<='1';
42             nx_state <= RY;
43             IF (test='0') THEN time <= timeRG;
44             ELSE time <= timeTEST;
45             END IF;
46         WHEN RY =>
47             r1<='1'; r2<='0'; y1<='0'; y2<='1'; g1<='0'; g2<='0';
48             nx_state <= GR;
49             IF (test='0') THEN time <= timeRY;
50             ELSE time <= timeTEST;
51             END IF;
52         WHEN GR =>
53             r1<='0'; r2<='1'; y1<='0'; y2<='0'; g1<='1'; g2<='0';
54             nx_state <= YR;
55             IF (test='0') THEN time <= timeGR;
56             ELSE time <= timeTEST;
57             END IF;
58         WHEN YR =>
59             r1<='0'; r2<='1'; y1<='1'; y2<='0'; g1<='0'; g2<='0';
60             nx_state <= RG;
61             IF (test='0') THEN time <= timeYR;
62             ELSE time <= timeTEST;
63             END IF;
64         WHEN YY =>
65             r1<='0'; r2<='0'; y1<='1'; y2<='1'; g1<='0'; g2<='0';
66             nx_state <= RY;
67     END CASE;
68 END PROCESS;
69 END behavior;
```


Assignments

- The assignments will be attached to your class room.

End of lecture 9

Any Questions ?